A Simple Spreadsheet Model to Incorporate Seasonal Growth into Length-Based Stock Assessment Methods

G.J. de Graaf and P. Dekker

Abstract

The paper describes a method by which seasonal growth can be incorporated into length-converted catch curves and cohort analyses using a spreadsheet. The method is based on calculating the length of fish using seasonal growth parameters on a daily basis. A LOOKUP function is then used to find the age corresponding to the length.

Introduction

Over the last decade, fisheries biologists working in tropical waters became increasingly familiar with 'length-based fish stock assessment', thanks to the development of easy-to-use software (such as LFSA, ELEFAN, FISAT, and LFDA); good training manuals (Sparre and Venema 1992, 1998); the worldwide FAO/DANIDA training course in Tropical Fish Stock Assessment; and the availability of relatively cheap computers.

Most of the traditional stock assessment methods work with age composition data, whereby annuli (on otoliths, scales and other bones) are used to estimate growth. However, in tropical waters this type of age reading is almost impossible and stock assessment became much easier with the development of length-based methods.

In principle, length-based fish stock assessment is based on the conversion of length into age, where it is assumed that the fish is growing according to the von Bertalanffy Growth Function (VBGF). It is generally accepted that in temperate waters the growth of fish has strong seasonal fluctuations, mainly due to fluctuations of temperature and/or food supply (Shul'man 1974). Strong seasonal fluctuations in growth also exist in tropical waters (Daget and Ecoutin 1976; de Graaf and Ofori-Danson 1997; de Graaf 2003). The use of a seasonal version of the VBGF has been discussed extensively by several authors (Pauly and Ingles 1981; Pauly et al. 1992; Longhurst and Pauly 1987; Sparre 1990) and seasonal versions of standard analytical models, such as the yield-per-recruit method (Sparre 1991) and length-converted catch curves (Pauly 1990), were developed and incorporated in the software packages. However, these models are very basic methods of stock assessment and provide only a first approximation of the status of fish stocks and the impact of fishing mortality. More detailed information is required for the formulation of fisheries management strategies. This is available through Virtual Population Analysis (VPA) or length-based cohort analysis and Thompson and Bell models (Sparre and Venema 1992). Pauly et al. (1987) used the seasonal version of the VBGF to slice the cohorts in a time-based VPA for Peruvian Anchoveta. This technique has been incorporated into the DOS version of FISAT, but the length-based VPA in FISAT is still non-seasonal. Therefore, the analysis of data with a spreadsheet is dependent on non-seasonal models that will give distorted results if the growth is seasonal.

This paper presents an approach in which seasonal growth can be incorporated in some major fish stock assessment methods using a spreadsheet.

Non Seasonal and Seasonal Growth

The basic tool for length-based methods is the conversion of length-based data into age-based data. Traditionally the VBGF is used for the conversion of length into age in length-based fish stock assessments. The non-seasonal version of VBGF takes the form:

$$L_t = L_\infty * \left[1 - e^{\left(-K*\left(t-t_0\right)\right)}\right] \qquad (1)$$

Where

L_t Length at time t
L_{oo} L infinitive or asymptotic length
K growth parameter
t_o T zero, or time when the fish are born or entered in the system

The growth rate at any point in the lifespan of the fish can be calculated as:

$$\frac{dL}{dt} = K\left(L_\infty - L_t\right) \qquad (2)$$

Conversion of length into age is done with the inverse VBGF:

$$t(L) = t_o - \frac{1}{K}\ln\left(1 - \frac{L}{L_\infty}\right) \qquad (3)$$

In conclusion, the conversion of length-based data into age-based data for growth without seasonality is rather straightforward. However, this is not the case where there are seasonal fluctuations in growth. The seasonal version of the VBGF (Somers 1988) has the following form:

$$L(t) = L_\infty\left\{1 - e^{-K(t-t_o)-(CK/2\pi)[\sin 2\pi(t-t_s)-\sin 2\pi(t_o-t_s)]}\right\} \qquad (4)$$

Where:

L_t Length at time t

L_∞ L infinitive

K growth rate parameter

t_0 T-zero

t_s the onset of the first oscillation relative to t = 0, or t_s = Winter point + 0.5"

C the intensity of the (sinusoid) growth oscillations

The parameter C is important as it determines the intensity of the seasonal growth. When $C = 0$, seasonal growth is absent and the equation equals the standard VBGF. At $C = 1$, growth is highly seasonal and comes to a standstill once a year at the winter-point. Intermediate values of C indicate growth reduction during the winter, but growth never completely stops.

The differences in growth and growth rate (dL/dt) for both versions of the VBGF are illustrated in Figure 1.

The growth rate (dL/dt) is the differential of the VBGF and is solved with mathematical software (Heck 2003) into the form:

$$\frac{dL}{dt} = -L_\infty\left(-K - CK\cos\left(2\pi\left(t-t_s\right)\right)\right)e^{\left[-K(t-t_o)-\frac{CK(\sin(2\pi(t-t_s))-\sin(2\pi(t_o-t_s)))}{2\pi}\right]} \qquad (5)$$

However, an inverse seasonal VBGF does not have a direct solution due to the fact that on the right side of the VBGF t is found two times in the exponential factor and can only be solved "numerically".

Estimating age when length is known in a seasonal VBGF

Converting length into age for the seasonal VBGF using a numerical mathematical approach can be done using the Solver function in Excel (Lleonart, pers. com). However, as the Solver function cannot be used in a multiple form, this does not lead to a simple and practical solution. Therefore, we looked at how we could approach the value of t for a given value of L as follows:

- Calculate the length of the fish with daily intervals for a set of given parameters of the seasonal VBGF.
- Search in this data set the length to be converted into age.
- Take in this data set the corresponding age value or t.

In a spreadsheet, this can be done with a LOOKUP function, as explained below.

Step 1. calculating length

Figure 2 presents the setup of a spreadsheet used to calculate length with seasonal growth. To facilitate building, the VBGF has been separated first into smaller blocks [$K(t-t_o)$; $CK/2\pi$; $\sin 2\pi(t-t_s)$; $\sin 2\pi(t-t_o)$] that



Figure 1. Example of non-seasonal and seasonal growth of Puntius sophore. Non-seasonal: Loo=13, K=1.3; Seasonal: Loo=13, K=1.3, C=1, ts=0.5.

are combined again in the column cells I10 :I16 to calculate *L(t)*. In this example from Bangladesh, the fish are born on May 15th (t_o=*0.37*) and have the highest growth in June and a reduced growth in December/January. In the column cells D10:D16 the age of the fish is calculated and together with the column cells I10:I16 they are the basic input for the conversion of length into age. Figure 3 presents the formulae for the different spreadsheet cells. Only the first few days' growth is presented in the example. With the Excel cell copying feature, the formulae in each relevant column can be copied down to row 1104 in order to cover a period of three years.

	A	B	C	D	E	F	G	H	I
1	*Lt=Linf*(1-exp(-K*(t-to)-(C*K/2pi)*(sin2pi(t-ts)-sin2pi(to-ts))*								
2	Parameters								
3	Linf	13							
4	K	1.3							
5	C	1							
6	ts	0.5	=Winterpoint+0.5						
7	to	0.370	15 may						
8									
9		Day	t	Age	-K(t-to)	CK/2pi	sin[2pi(t-ts)]	sin[2pi(to-ts)]	L(t)
10	start at "to"	0	0.370	0	0.000	0.207	-0.729	-0.729	0.00
11		1	0.373	0	-0.004	0.207	-0.717	-0.729	0.08
12		2	0.375	0.01	-0.007	0.207	-0.705	-0.729	0.16
13		3	0.378	0.01	-0.011	0.207	-0.693	-0.729	0.23
14		4	0.381	0.01	-0.014	0.207	-0.680	-0.729	0.31
15		5	0.384	0.01	-0.018	0.207	-0.668	-0.729	0.39
16		6	0.386	0.02	-0.021	0.207	-0.655	-0.729	0.47

Figure 2. Calculating length in a spreadsheet.[1]

	A	B	C	D	E	F	G	H	I
1	*Lt=Linf*(1-exp(-K*(t-to)-(C*K/2pi)*(sin2pi(t-ts)-sin2pi(to-ts))*								
2	Parameters								
3	Linf	13							
4	K	1.3							
5	C	1							
6	ts	0.5	=Winterpoint+0.5						
7	to	0.37	15 May						
8									
9		Day	t	Age	-K(t-to)	CK/2pi	sin[2pi(t-ts)]	sin[2pi(to-ts)]	L(t)
10		0	=B7	=C10-$B$7	=-1*$B$4*(C10-$B$7)	=($B$5*$B$4)/(2*3.14)	=SIN(2*3.14*(C10-$B$6))	=SIN(2*3.14*($B$7-$B$6))	=$B$3*(1-EXP(E10-F10*(G10-H10)))
11		=B10+1	=C10+(1/365)	=C11-$B$7	=-1*$B$4*(C11-$B$7)	=($B$5*$B$4)/(2*3.14)	=SIN(2*3.14*(C11-$B$6))	=SIN(2*3.14*($B$7-$B$6))	=$B$3*(1-EXP(E11-F11*(G11-H11)))
12		=B11+1	=C11+(1/365)	=C12-$B$7	=-1*$B$4*(C12-$B$7)	=($B$5*$B$4)/(2*3.14)	=SIN(2*3.14*(C12-$B$6))	=SIN(2*3.14*($B$7-$B$6))	=$B$3*(1-EXP(E12-F12*(G12-H12)))
13		=B12+1	=C12+(1/365)	=C13-$B$7	=-1*$B$4*(C13-$B$7)	=($B$5*$B$4)/(2*3.14)	=SIN(2*3.14*(C13-$B$6))	=SIN(2*3.14*($B$7-$B$6))	=$B$3*(1-EXP(E13-F13*(G13-H13)))
14		=B13+1	=C13+(1/365)	=C14-$B$7	=-1*$B$4*(C14-$B$7)	=($B$5*$B$4)/(2*3.14)	=SIN(2*3.14*(C14-$B$6))	=SIN(2*3.14*($B$7-$B$6))	=$B$3*(1-EXP(E14-F14*(G14-H14)))
15		=B14+1	=C14+(1/365)	=C15-$B$7	=-1*$B$4*(C15-$B$7)	=($B$5*$B$4)/(2*3.14)	=SIN(2*3.14*(C15-$B$6))	=SIN(2*3.14*($B$7-$B$6))	=$B$3*(1-EXP(E15-F15*(G15-H15)))
16		=B15+1	=C15+(1/365)	=C16-$B$7	=-1*$B$4*(C16-$B$7)	=($B$5*$B$4)/(2*3.14)	=SIN(2*3.14*(C16-$B$6))	=SIN(2*3.14*($B$7-$B$6))	=$B$3*(1-EXP(E16-F16*(G16-H16)))

Figure 3. The cell formulae to calculate length with seasonal growth.

Step 2. inserting the lookup function

With the LOOKUP function in Excel, the instruction is given to search for an indicated value in a column or row and, once the value is found, to provide a corresponding value from another column or row. In our case, LOOKUP searches for a given length in the column where the calculated length data are stored and returns the corresponding age for this length from the column where the ages are stored. The LOOKUP function has three entries:

1. Lookup_value is a value that LOOKUP searches for in Lookup_vector and can be a number, a text, or a logical value. In our case it is the *L(t)*.
2. Lookup_vector is a range that contains only one row or one column of numbers, text or logical values, placed in ascending order. In our case it is the column with the 'length data'.
3. Result_vector is a range that contains only one row or column of numbers, text or logical values and has the same size as the Lookup_vector. In our case it is the column with the 'age data'.

An example of conversion of length into age with the LOOKUP function is presented in Figure 4. In cells M10: M22 and N10:N22 the length classes are entered with a 1 cm interval. The mid-length of each length class is calculated in the cells O10:O22. With the LOOKUP function entered in cells P10:P22, the age of each mid-length is estimated in the spreadsheet. In cell O10, the mid-length of the interval 0-1 cm is calculated (0.5 cm), and with the LOOKUP function the column *L* is searched for the first value approaching 0.5. Once found, in cell L16, it takes the corresponding age value from column K and enters

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P
1	*Lt=Linf*(1-exp(-K*(t-to)-(C*K/2pi)*(sin2pi(t-ts)-sin2pi(to-ts))*															
2	Parameters for *Puntius sophore*, Bangladesh															
3	Linf	13.0														
4	K	1.3														
5	C	1.0														
6	ts	0.50	=Winterpoint+0.5								CONVERTING LENGTH INTO AGE					
7	to	0.37	15 may													
8																With LOOKUP
9		Day	t	Age	-K(t-to)	CK/2pi	in[2pi(t-ts	[2pi(to-t	L(t)		Age	L(t)	L1	L2	Mid length	Age of mid length
10	start	0.000	0.370	0.000	0.000	0.207	-0.729	-0.729	0.000		0.000	0.000	0.0	1.0	0.5	*0.016*
11		1.000	0.373	0.003	-0.004	0.207	-0.717	-0.729	0.078		0.003	0.078	1.0	2.0	1.5	*0.052*
12		2.000	0.375	0.005	-0.007	0.207	-0.705	-0.729	0.156		0.005	0.156	2.0	3.0	2.5	*0.088*
13		3.000	0.378	0.008	-0.011	0.207	-0.693	-0.729	0.234		0.008	0.234	3.0	4.0	3.5	*0.126*
14		4.000	0.381	0.011	-0.014	0.207	-0.680	-0.729	0.313		0.011	0.313	4.0	5.0	4.5	*0.170*
15		5.000	0.384	0.014	-0.018	0.207	-0.668	-0.729	0.391		0.014	0.391	5.0	6.0	5.5	*0.219*
16		6.000	0.386	0.016	-0.021	0.207	-0.655	-0.729	0.469		0.016	0.469	6.0	7.0	6.5	*0.285*
17		7.000	0.389	0.019	-0.025	0.207	-0.642	-0.729	0.548		0.019	0.548	7.0	8.0	7.5	*0.386*
18		8.000	0.392	0.022	-0.028	0.207	-0.628	-0.729	0.626		0.022	0.626	8.0	9.0	8.5	*0.855*
19		9.000	0.395	0.025	-0.032	0.207	-0.615	-0.729	0.704		0.025	0.704	9.0	10.0	9.5	*1.005*
20		10.000	0.397	0.027	-0.036	0.207	-0.601	-0.729	0.783		0.027	0.783	10.0	11.0	10.5	*1.140*
21		11.000	0.400	0.030	-0.039	0.207	-0.588	-0.729	0.861		0.030	0.861	11.0	12.0	11.5	*1.384*

Figure 4. Converting length into age with the LOOKUP function.

[1] In the formula in the spreadsheet t_s or the onset of the first oscillation relative to *t = 0* is used. Users of FISAT are more familiar with the Winterpoint WP = t_s+*0.5*

this value in cell P10. In our example the first value, approaching 0.5 cm, is 0.469 cm on day 6, with an age of 0.016 years. It could be argued that the method is rather inaccurate. However, in the example, the fish are growing with steps of one day. Reducing this to steps of 0.25 day improves the accuracy considerably but also increases the size of the spreadsheet.

The formulae for the different cells are presented in Figure 5.

Length-based methods are often used when length composition data for the total fishery is available for a one year period only. This is often the case in fish stock assessment programs in developing countries where the funds for continuous monitoring of the fishery are lacking. Catch curves and cohort analysis can be applied under these conditions as their basic assumption is the constant parameter system. The picture presented by all length classes caught during one year reflects that of a single cohort during its entire life span (Sparre and Venema 1992).

Linearized Length-converted Catch Curve

The construction of a catch curve is the most common approach to estimate the total mortality of a cohort. Details of the method are well presented by Sparre and Venema (1992) and Gayanilo and Pauly (1997) and are only briefly summarized here. Assuming constant recruitment and constant mortality the length converted catch curves takes the form:

$$\ln\left(\frac{C_i}{dt_i}\right) = a + Zt_i' \tag{6}$$

Where

- C catch number of length class i
- dt_i time needed for the fish to grow through length class i
- Z total mortality
- t_i' age of the mid-length of length class i
- a constant

For non-seasonal growth dt_i is estimated from:

$$dt_i = \left(\frac{1}{K}\right)\ln\left[\frac{L_\infty - L_{i+1}}{L_\infty - L_i}\right]$$

and t_i' is estimated from:

$$t_i' = \left(\frac{1}{K}\right)\ln\left[1-\left(\frac{L_i}{L_\infty}\right)\right] \tag{7}$$

Sparre (1990) and Pauly (1990) clearly demonstrated that the total mortality is overestimated by the traditional catch curve if seasonal growth is not accounted for. The reason is that dt_i and t_i' depend not only on length but also on the time of the year if growth is seasonal. Pauly (1990) developed a method using the parameters of the seasonal VBGF to identify a number of pseudo cohorts to resolve this problem. This method is incorporated into FISAT. As it is rather complicated to apply this method in a spreadsheet, we explain how a catch curve can be made with the LOOKUP function to provide results comparable to those with Pauly's method. We illustrate this with an example of *Puntius sophore* from Bangladesh.

The number of *Puntius sophore* caught in a one-year stock assessment program for the different length classes is presented in Table 1.

	K	L	M	N	O	P
6	CONVERTING LENGTH INTO AGE					
7						
8						With LOOKUP
9	=D9	=I9	L1	L2	Mid length	Age of mid length
10	=D10	=I10	0	1	=(M10+N10)/2	=LOOKUP(O10,L$10:L$1104,K$10:K$1104)
11	=D11	=I11	=N10	=M11+1	=(M11+N11)/2	=LOOKUP(O11,L$10:L$1104,K$10:K$1104)
12	=D12	=I12	=N11	=M12+1	=(M12+N12)/2	=LOOKUP(O12,L$10:L$1104,K$10:K$1104)
13	=D13	=I13	=N12	=M13+1	=(M13+N13)/2	=LOOKUP(O13,L$10:L$1104,K$10:K$1104)
14	=D14	=I14	=N13	=M14+1	=(M14+N14)/2	=LOOKUP(O14,L$10:L$1104,K$10:K$1104)
15	=D15	=I15	=N14	=M15+1	=(M15+N15)/2	=LOOKUP(O15,L$10:L$1104,K$10:K$1104)
16	=D16	=I16	=N15	=M16+1	=(M16+N16)/2	=LOOKUP(O16,L$10:L$1104,K$10:K$1104)
17	=D17	=I17	=N16	=M17+1	=(M17+N17)/2	=LOOKUP(O17,L$10:L$1104,K$10:K$1104)
18	=D18	=I18	=N17	=M18+1	=(M18+N18)/2	=LOOKUP(O18,L$10:L$1104,K$10:K$1104)
19	=D19	=I19	=N18	=M19+1	=(M19+N19)/2	=LOOKUP(O19,L$10:L$1104,K$10:K$1104)
20	=D20	=I20	=N19	=M20+1	=(M20+N20)/2	=LOOKUP(O20,L$10:L$1104,K$10:K$1104)
21	=D21	=I21	=N20	=M21+1	=(M21+N21)/2	=LOOKUP(O21,L$10:L$1104,K$10:K$1104)
22	=D22	=I22	=N21	=M22+1	=(M22+N22)/2	=LOOKUP(O22,L$10:L$1104,K$10:K$1104)
23	=D23	=I23				

Figure 5. The cell formulae for the conversion of length into age with a LOOKUP function.

Table 1. Length frequency distribution and estimated growth parameters of ***Puntius sophore*** as obtained through a one-year stock assessment program.

L1 (cm)	L2 (cm)	Mid length (cm)	Catch numbers	Growth para-meters	
0.0	1.0	0.5	10	Linf	13
1.0	2.0	1.5	250	K	1.3
2.0	3.0	2.5	590	*to*	0.37
3.0	4.0	3.5	70	ts	0.5
4.0	5.0	4.5	520	C	1
5.0	6.0	5.5	2160		
6.0	7.0	6.5	3830		
7.0	8.0	7.5	1970		
8.0	9.0	8.5	1150		
9.0	10.0	9.5	490		
10.0	11.0	10.5	270		
11.0	12.0	11.5	50		
12.0	13.0	12.5	30		

The distribution is believed to be representative for the overall population structure as the fish were caught with non-selective gears.

We have to calculate dt_i, the time needed for the fish to grow through length class l in order to construct the catch curve. For each length class this is done with the LOOKUP function as follows:

$$dt = \text{age } L2 - \text{age } L1 \qquad (8)$$

Conversion of the mid-length into age has been demonstrated in the previous paragraph and construction of the catch curve then becomes straightforward (Figures 6 and 7).

In column S, delta t (dt) is calculated and ln(C/dt) is calculated in column T. Plotting Ln(C/dt) against the age of the mid-length (column P) gives the catch curve (Figure 8) indicating a total mortality of 4.57 year^{-1}, which is similar to the results of analysing the data in FISAT with the method of Pauly (1990).

	K	L	M	N	O	P	Q	R	S	T	U	V
6						LENGTH CONVERTED CATCH CURVE						
7												
8						lookup	lookup	lookup				
9	Age	L(t)	L1	L2	Mid length	Age of mid length	Age L1	Age L2	dt	Catch numbers	Ln(C/dt)	Used
10	0.000	~~0.000~~	~~0.0~~	~~1.0~~	~~0.5~~	~~**0.016**~~	0.000	0.033	0.033	10	5.7	
11	0.003	0.078	1.0	2.0	1.5	**0.052**	0.033	0.068	0.036	250	8.9	
12	0.005	0.156	2.0	3.0	2.5	**0.088**	0.068	0.107	0.038	590	9.6	
13	0.008	0.234	3.0	4.0	3.5	**0.126**	0.107	0.148	0.041	70	7.4	
14	0.011	0.313	4.0	5.0	4.5	**0.170**	0.148	0.195	0.047	520	9.3	
15	0.014	0.391	5.0	6.0	5.5	**0.219**	0.195	0.249	0.055	2160	10.6	10.6
16	0.016	0.469	6.0	7.0	6.5	**0.285**	0.249	0.326	0.077	3830	10.8	10.8
17	0.019	0.548	7.0	8.0	7.5	**0.386**	0.326	0.507	0.181	1970	9.3	9.3
18	0.022	0.626	8.0	9.0	8.5	**0.855**	0.507	0.937	0.430	1150	7.9	7.9
19	0.025	0.704	9.0	10.0	9.5	**1.005**	0.937	1.068	0.132	490	8.2	8.2
20	0.027	0.783	10.0	11.0	10.5	**1.140**	1.068	1.230	0.162	270	7.4	7.4
21	0.030	0.861	11.0	12.0	11.5	**1.384**	1.230	1.984	0.753	50	4.2	4.2
22	0.033	0.939	12.0	13.0	12.5	**2.268**	1.984	2.997	1.014	30	3.4	
23	0.036	1.017										

Figure 6. Constructing a seasonalized catch curve.

Cohort or Virtual Population Analysis[2]

Cohort or Virtual Population Analysis use the number of fish caught during commercial fishing operations to estimate historic fishing mortality

	K	L	M	N	O	P	Q	R	S	T	U	V
6						LENGTH CONVERTED CATCH CURVE						
7												
8						lookup	lookup	lookup				
9	Age	L(t)	L1	L2	Mid length	Age of mid length	Age L1	Age L2	dt	Catch numbers	Ln(C/dt)	Used
10	=D10	=I10	0	1	=(M10+N10)/2	**=LOOKUP(O10,L$10:L$1104,K$10:K$1104)**	=LOOKUP(M10,L$10:L$1104,K$10:K$1104)	=LOOKUP(N10,L$10:L$1104,K$10:K$1104)	=R10-Q10	10	=LN(T10/S10)	
11	=D11	=I11	=N10	=M11+1	=(M11+N11)/2	**=LOOKUP(O11,L$10:L$1104,K$10:K$1104)**	=LOOKUP(M11,L$10:L$1104,K$10:K$1104)	=LOOKUP(N11,L$10:L$1104,K$10:K$1104)	=R11-Q11	250	=LN(T11/S11)	
12	=D12	=I12	=N11	=M12+1	=(M12+N12)/2	**=LOOKUP(O12,L$10:L$1104,K$10:K$1104)**	=LOOKUP(M12,L$10:L$1104,K$10:K$1104)	=LOOKUP(N12,L$10:L$1104,K$10:K$1104)	=R12-Q12	590	=LN(T12/S12)	
13	=D13	=I13	=N12	=M13+1	=(M13+N13)/2	**=LOOKUP(O13,L$10:L$1104,K$10:K$1104)**	=LOOKUP(M13,L$10:L$1104,K$10:K$1104)	=LOOKUP(N13,L$10:L$1104,K$10:K$1104)	=R13-Q13	70	=LN(T13/S13)	
14	=D14	=I14	=N13	=M14+1	=(M14+N14)/2	**=LOOKUP(O14,L$10:L$1104,K$10:K$1104)**	=LOOKUP(M14,L$10:L$1104,K$10:K$1104)	=LOOKUP(N14,L$10:L$1104,K$10:K$1104)	=R14-Q14	520	=LN(T14/S14)	
15	=D15	=I15	=N14	=M15+1	=(M15+N15)/2	**=LOOKUP(O15,L$10:L$1104,K$10:K$1104)**	=LOOKUP(M15,L$10:L$1104,K$10:K$1104)	=LOOKUP(N15,L$10:L$1104,K$10:K$1104)	=R15-Q15	2160	=LN(T15/S15)	=U15
16	=D16	=I16	=N15	=M16+1	=(M16+N16)/2	**=LOOKUP(O16,L$10:L$1104,K$10:K$1104)**	=LOOKUP(M16,L$10:L$1104,K$10:K$1104)	=LOOKUP(N16,L$10:L$1104,K$10:K$1104)	=R16-Q16	3830	=LN(T16/S16)	=U16
17	=D17	=I17	=N16	=M17+1	=(M17+N17)/2	**=LOOKUP(O17,L$10:L$1104,K$10:K$1104)**	=LOOKUP(M17,L$10:L$1104,K$10:K$1104)	=LOOKUP(N17,L$10:L$1104,K$10:K$1104)	=R17-Q17	1970	=LN(T17/S17)	=U17
18	=D18	=I18	=N17	=M18+1	=(M18+N18)/2	**=LOOKUP(O18,L$10:L$1104,K$10:K$1104)**	=LOOKUP(M18,L$10:L$1104,K$10:K$1104)	=LOOKUP(N18,L$10:L$1104,K$10:K$1104)	=R18-Q18	1150	=LN(T18/S18)	=U18
19	=D19	=I19	=N18	=M19+1	=(M19+N19)/2	**=LOOKUP(O19,L$10:L$1104,K$10:K$1104)**	=LOOKUP(M19,L$10:L$1104,K$10:K$1104)	=LOOKUP(N19,L$10:L$1104,K$10:K$1104)	=R19-Q19	490	=LN(T19/S19)	=U19
20	=D20	=I20	=N19	=M20+1	=(M20+N20)/2	**=LOOKUP(O20,L$10:L$1104,K$10:K$1104)**	=LOOKUP(M20,L$10:L$1104,K$10:K$1104)	=LOOKUP(N20,L$10:L$1104,K$10:K$1104)	=R20-Q20	270	=LN(T20/S20)	=U20
21	=D21	=I21	=N20	=M21+1	=(M21+N21)/2	**=LOOKUP(O21,L$10:L$1104,K$10:K$1104)**	=LOOKUP(M21,L$10:L$1104,K$10:K$1104)	=LOOKUP(N21,L$10:L$1104,K$10:K$1104)	=R21-Q21	50	=LN(T21/S21)	=U21
22	=D22	=I22	=N21	=M22+1	=(M22+N22)/2	**=LOOKUP(O22,L$10:L$1104,K$10:K$1104)**	=LOOKUP(M22,L$10:L$1104,K$10:K$1104)	=LOOKUP(N22,L$10:L$1104,K$10:K$1104)	=R22-Q22	30	=LN(T22/S22)	

Figure 7. The cell formulae for the seasonalized length-converted catch curve.



Figure 8. Results of length-converted catch curve of *Puntius sophore* made in a spreadsheet (A) and with FISAT (B).

[2] Adapted from Gayanilo and Pauly (1997) and King (1995).

and stock numbers in a cohort of fish and is again based on the constant parameter system.

The number of fish surviving from one year (N_t) to the next year (N_{t+1}) is given by:

$$N_{t+1} = N_t * e^{[-(F_t+M)]} \quad (9)$$

The number of dying fish is therefore:

$$N_t * \left(1 - e^{-Z}\right) \quad (10)$$

The catch (C_t) is the proportion dying owing to fishing, and may be estimated from the catch or the Baranov (1926) equation;

$$C_t = \left[{F_t}/{Z}\right] * N_t * \left(1 - e^{[-(F+M)]}\right) \quad (11)$$

Combining the different equations will give the Gulland (1965) equation for Virtual Population Analysis:

$$C_t/N_{t+1} = \left(F_t/Z_t\right) * \left(e^{Z_t} - 1\right) \quad (12)$$

Given values of the catch (C_t), and an estimate of the natural mortality M, the equation can be used to estimate retroactively the size of the past cohorts, if an estimate of N_{t+1} is available from which to start the computation. Estimates of N_{t+1} (expressing the last population size a cohort had before it became extinct) are called 'terminal population' (N_t). Values of N_t can be obtained from:

$$N_t = Z_t * C_t/F_t \quad (13)$$

Where C_t is the terminal catch (i.e., the last catch taken from a cohort before it went extinct) and F_t is the terminal fishing mortality (i.e., the fishing pressure that generated C_t). A VPA starts with an initial guess of F_t and then calculates backwards with the known catches and natural mortality rate (Figure 9).



Figure 9. The principle of backward calculation in a VPA (King 1995).

The major aim of cohort analysis is to estimate the fishing mortality (F) over the different length classes. The basics on how to do it are well explained by Sparre and Venema (1992, 1998) and only summarized here.

The basic formulae for a length based cohort analysis are:

$$N(L1) = \left[N(L2).e^{\left(\frac{M.dt}{2}\right)} + C(L1,L2)\right].e^{\left(\frac{M.dt}{2}\right)}$$

and

$$C(L1,L2) = N(L1)\frac{F}{Z}\left[1 - e^{-Z.dt}\right] \quad (14)$$

Where

$C(L1,L2)$ the number of fish caught of length between $L1$ and $L2$
$N(L1)$ the number of fish that attain length $L1$
dt time needed for the fish to grow through length $L1$ to length $L2$
M natural mortality during time dt
F fishing mortality during time dt
Z total mortality during time dt

In a cohort analyses with non-seasonal growth $e^{\left(\frac{M.dt}{2}\right)}$, the fraction of $N(L1)$ that survives natural death during the time period from $t(L1)$ to $t(L2)$ and dt is calculated as:

$$e^{\left(\frac{M.dt}{2}\right)} = \left[\frac{L\infty - L1}{L\infty - L2}\right]^{\frac{M}{2K}}$$

and

$$dt = \left(1/K\right)\ln\left[\frac{L\infty - L1}{L\infty - L2}\right] \quad (15)$$

Again with seasonal growth the last formulae will give incorrect results. However, dt can be calculated as: dt = age $L2$ - age $L1$ and can be solved with the LOOKUP function. Then we can use the basic formulae $e^{\left(\frac{M.dt}{2}\right)}$ directly.

In Figure 10 an example of a cohort analysis in a spreadsheet for *Puntius sophore* with a natural mortality of M = 1.168 year^{-1} (all other parameters being the same as those in the previous examples) is presented. The cell formulae are presented in Figure 11.

	K	L	M	N	O	P	Q	R	S	T	U	V	W
6	COHORT ANALYSIS												
7	Natural mortality (M)		1.168										
8						Look up	Look up						
9	Age	L(t)	L1	L2	mid length	Age L1	Age L2	dt	Catch numbers	exp(M*dT/2)	NL1=(N2*exp(M*dT/2)+Catch numbers)*exp(M*dT/2)	F/Z=Catch numbers/(N1-N2)	F=M*F/Z/(1-F/Z)
10	0.000	0.000	0	1	0.5	0.000	0.033	0.033	0	1.019	2765999	0.00	0.00
11	0.003	0.078	1	2	1.5	0.033	0.068	0.036	0	1.021	2653294	0.00	0.00
12	0.005	0.156	2	3	2.5	0.068	0.107	0.038	0	1.023	2537050	0.00	0.00
13	0.008	0.234	3	4	3.5	0.107	0.148	0.041	33614	1.024	2418148	0.21	0.31
14	0.011	0.313	4	5	4.5	0.148	0.195	0.047	154485	1.028	2257403	0.53	1.34
15	0.014	0.391	5	6	5.5	0.195	0.249	0.055	285728	1.033	1967835	0.65	2.14
16	0.016	0.469	6	7	6.5	0.249	0.326	0.077	330862	1.046	1525976	0.56	1.50
17	0.019	0.548	7	8	7.5	0.326	0.507	0.181	513255	1.111	937744	0.67	2.34
18	0.022	0.626	8	9	8.5	0.507	0.937	0.430	121710	1.286	168163	0.89	9.51
19	0.025	0.704	9	10	9.5	0.937	1.068	0.132	18618	1.080	31506	0.83	5.81
20	0.027	0.783	10	11	10.5	1.068	1.230	0.162	2595	1.099	9145	0.37	0.68
21	0.030	0.861	11	12	11.5	1.230	1.984	0.753	393	1.553	2122	0.23	0.35
22	0.033	0.939	12	13	12.5	1.984	2.997	1.014	216	1.808	432	0.50	1.17
23	0.036	1.017										Initial guess	

Figure 10. Example of cohort analysis with seasonal growth in a spreadsheet.

	K	L	M	N	O	P	Q	R	S	T	U	V	W
6	COHORT ANALYSIS												
7	Natural mortality (M)		1.168										
8						Look up	Look up						
9	Age	L(t)	L1	L2	mid length	Age L1	Age L2	dt	Catch numbers	exp(M*dT/2)	NL1=(N2*exp(M*dT/2)+Catch numbers)*exp(M*dT/2)	F/Z=Catch numbers/(N1-N2)	F=M*F/Z/(1-F/Z)
10	=D10	=I10	0	1	=(M10+N10)/2	=LOOKUP(M10,L$10:L$1104,K$10:K$1104)	=LOOKUP(N10,L$10:L$1104,K$10:K$1104)	=Q10-P10	0	=EXP(($M$7*R10)/2)	=(U11*T11+S10)*T11	=S10/(U10-U11)	=$M$7*V10/(1-V10)
11	=D11	=I11	=N10	=M11+1	=(M11+N11)/2	=LOOKUP(M11,L$10:L$1104,K$10:K$1104)	=LOOKUP(N11,L$10:L$1104,K$10:K$1104)	=Q11-P11	0	=EXP(($M$7*R11)/2)	=(U12*T12+S11)*T12	=S11/(U11-U12)	=$M$7*V11/(1-V11)
12	=D12	=I12	=N11	=M12+1	=(M12+N12)/2	=LOOKUP(M12,L$10:L$1104,K$10:K$1104)	=LOOKUP(N12,L$10:L$1104,K$10:K$1104)	=Q12-P12	0	=EXP(($M$7*R12)/2)	=(U13*T13+S12)*T13	=S12/(U12-U13)	=$M$7*V12/(1-V12)
13	=D13	=I13	=N12	=M13+1	=(M13+N13)/2	=LOOKUP(M13,L$10:L$1104,K$10:K$1104)	=LOOKUP(N13,L$10:L$1104,K$10:K$1104)	=Q13-P13	33614	=EXP(($M$7*R13)/2)	=(U14*T14+S13)*T14	=S13/(U13-U14)	=$M$7*V13/(1-V13)
14	=D14	=I14	=N13	=M14+1	=(M14+N14)/2	=LOOKUP(M14,L$10:L$1104,K$10:K$1104)	=LOOKUP(N14,L$10:L$1104,K$10:K$1104)	=Q14-P14	154485	=EXP(($M$7*R14)/2)	=(U15*T15+S14)*T15	=S14/(U14-U15)	=$M$7*V14/(1-V14)
15	=D15	=I15	=N14	=M15+1	=(M15+N15)/2	=LOOKUP(M15,L$10:L$1104,K$10:K$1104)	=LOOKUP(N15,L$10:L$1104,K$10:K$1104)	=Q15-P15	285728	=EXP(($M$7*R15)/2)	=(U16*T16+S15)*T16	=S15/(U15-U16)	=$M$7*V15/(1-V15)
16	=D16	=I16	=N15	=M16+1	=(M16+N16)/2	=LOOKUP(M16,L$10:L$1104,K$10:K$1104)	=LOOKUP(N16,L$10:L$1104,K$10:K$1104)	=Q16-P16	330862	=EXP(($M$7*R16)/2)	=(U17*T17+S16)*T17	=S16/(U16-U17)	=$M$7*V16/(1-V16)
17	=D17	=I17	=N16	=M17+1	=(M17+N17)/2	=LOOKUP(M17,L$10:L$1104,K$10:K$1104)	=LOOKUP(N17,L$10:L$1104,K$10:K$1104)	=Q17-P17	513255	=EXP(($M$7*R17)/2)	=(U18*T18+S17)*T18	=S17/(U17-U18)	=$M$7*V17/(1-V17)
18	=D18	=I18	=N17	=M18+1	=(M18+N18)/2	=LOOKUP(M18,L$10:L$1104,K$10:K$1104)	=LOOKUP(N18,L$10:L$1104,K$10:K$1104)	=Q18-P18	121710	=EXP(($M$7*R18)/2)	=(U19*T19+S18)*T19	=S18/(U18-U19)	=$M$7*V18/(1-V18)
19	=D19	=I19	=N18	=M19+1	=(M19+N19)/2	=LOOKUP(M19,L$10:L$1104,K$10:K$1104)	=LOOKUP(N19,L$10:L$1104,K$10:K$1104)	=Q19-P19	18618	=EXP(($M$7*R19)/2)	=(U20*T20+S19)*T20	=S19/(U19-U20)	=$M$7*V19/(1-V19)
20	=D20	=I20	=N19	=M20+1	=(M20+N20)/2	=LOOKUP(M20,L$10:L$1104,K$10:K$1104)	=LOOKUP(N20,L$10:L$1104,K$10:K$1104)	=Q20-P20	2595	=EXP(($M$7*R20)/2)	=(U21*T21+S20)*T21	=S20/(U20-U21)	=$M$7*V20/(1-V20)
21	=D21	=I21	=N20	=M21+1	=(M21+N21)/2	=LOOKUP(M21,L$10:L$1104,K$10:K$1104)	=LOOKUP(N21,L$10:L$1104,K$10:K$1104)	=Q21-P21	393	=EXP(($M$7*R21)/2)	=(U22*T22+S21)*T22	=S21/(U21-U22)	=$M$7*V21/(1-V21)
22	=D22	=I22	=N21	=M22+1	=(M22+N22)/2	=LOOKUP(M22,L$10:L$1104,K$10:K$1104)	=LOOKUP(N22,L$10:L$1104,K$10:K$1104)	=Q22-P22	216	=EXP(($M$7*R22)/2)	=S22/V22	0.5	=$M$7*V22/(1-V22)
23	=D23	=I23										Initial guess	

Figure 11. Cell formulae for cohort analysis with seasonal growth.

In Figure 12 the fishing mortality (F) for the different length classes, calculated with non-seasonal and seasonal growth, is presented. The comparison indicates that the traditional non-seasonal method underestimates F during the period of slow growth, which in our example is occurring at a length of 8-9 cm. The main reason is that the fish stay for a long period in this length class, while fishing continues. The difference seems to be subtle, but the practical consequences can be large. For example, in Bangladesh, growth slows during the dry season (December-January). During this period the floodplains are small as they are drying up and the fishing effort is high. Underestimating the fishing mortality during this period will have serious consequences if the estimates are used in the Thompson and Bell models for the comparison of fisheries management options.

In this respect, it is important to notice that the LOOKUP function can be used in a similar way to calculate *dt* for seasonal growth in the Thompson and Bell models and in the seasonal version of the Yield per Recruit Analyses (Sparre 1991).

Limitation of the Proposed Method

The proposed method provides convenient results but has some limitations. First of all, the time of recruitment has to be known. This is a minor limitation as, in most cases, the major spawning month is known.

Secondly, the used seasonal version of the VBGF of Somers (1988) has exactly one zero growth rate per year when C = 1, which means that for each length there will always be one value for age. Using this version of the VBGF will do for most tropical fisheries, where prolonged periods of zero growth are an exception. The method cannot be applied if the



Figure 12. Results of a cohort analysis with seasonal growth and the fishing mortality as estimated by using the seasonal or non-seasonal VBGF.

seasonal version of the VBGF of Pauly et al. (1992) is used, as this version allows for longer periods of 'no growth', which means that a length can have several values for age and we cannot convert length into age.

A similar problem arises if there are two cohorts per year, which is often the case in penaeid shrimps. There again, there is no one-to-one correspondence between age and length (Sparre 1990). In this case, the only solution is to slice the cohorts (Sparre and Venema 1998) and apply a VPA with pseudo cohorts (Pauly et al. 1987; Gayanilo and Pauly 1997).

The Spreadsheets

The different spreadsheets can be downloaded from our website www.nefisco.org/Training.htm

Acknowledgements

We would like to thank Mr Jordi Lleonart (FAO, Rome, Italy) and Mr. Daniel Pauly (University of British Colombia, Vancouver, Canada) for their useful comments on the first version of the manuscript and, of course, Tracy Barnett for the editing of the document.

References

Baranov, F.I. 1926. On the question of the dynamics of the fishing industry. Naunchn Byull. Rybn. Khoz. 8(1925):7-11 (in Russian).

Daget, J. and J.M. Ecoutin. 1976. Modeles mathematiques de production applicables aux poissons tropicaux subissant un arret annuel prolonge de croissance. Cahiers ORSTOM, serie hydrobiologie 10(2):59-69.

de Graaf, G.J. and P.K. Ofori-Danson. 1997. Catch and fish stock assessment in stratum VII of Lake Volta. IDAF Technical Report/97/1, GHA/93/008. FAO. Rome.

de Graaf, G.J. 2003. The floodpulse and growth of floodplain fish in Bangladesh. Fisheries Management and Ecology (10):241-247.

Gayanilo, F.C. and D. Pauly. 1997. FAO-ICLARM stock assessment tools (FISAT), reference manual. FAO computerized information series (Fisheries) No. 8. FAO, Rome.

Gulland, J.A. 1965. Estimation of mortality rates. Annex to Artic fisheries working group report ICES C.M. 1965/D:3. p. 231-241. In: P.H. Cushing (ed.) Key papers on fish populations. IRL Press, Oxford, 1983.

Heck, A. 2003. Introduction to Maple. 3rd Edition. Springer-Verlag, New York.

King, M. 1995. Fisheries Biology, Assessment and Management. Fishing News Books, Oxford, UK.

Longhurst, A.R. and D. Pauly. 1987. Ecology of Tropical Oceans. Academic Press, San Diego, California.

Pauly, D. 1990. Length converted catch curves and the seasonal growth of fishes. ICLARM Fishbyte 8(3):33-38.

Pauly, D. and J. Ingles. 1981. Aspects of growth and natural mortality of exploited coral reef fishes. p. 89-98. In: E. Gomez, C.E. Birkeland, R.W. Buddemeyer, R.E. Johannes, J.A. Marsh Jr. and R.T. Tsuda (eds.) Proceedings of the

Fourth International Coral Reef Symposium, Manila, Philippines, Vol. 1.
Pauly, D., M.L. Palomares and F.C. Gayanilo. 1987. VPA estimates of monthly population length composition, recruitment, mortality, biomass and related statistics of Peruvian Anchoveta, 1953 to 1981. p. 142-166. In: D. Pauly and I. Tsukayama (eds.) ICLARM Stud. Rev. 15.
Pauly, D., M. Soriano-Bartz, J. Moreau and A. Jarre Teichman. 1992. A new model accounting for seasonal cessation of growth in fishes. Australian Journal Marine and Freshwater Research 43.
Shul'man G.E. 1974. Life Cycles of Fish: Physiology and Biochemistry. Wiley and Sons, New York.
Somers I.F. 1988. On a seasonal oscillating growth function. ICLARM Fishbyte 6(1):8-11.
Sparre, P. 1990. Can we use traditional length-based fish stock assessment when growth is seasonal. ICLARM Fishbyte 8(3):29-32.
Sparre, P. 1991. Estimation of yield per recruit when growth and fishing mortality oscillate seasonally. ICLARM Fishbyte 9(1):40-44.
Sparre, P. and S.C. Venema. 1992. Introduction to tropical fish stock assessment, Part 1-manual. FAO Fisheries technical paper 306-1, rev. 1.
Sparre, P. and S.C. Venema. 1998. Introduction to tropical fish stock assessment, Part 1-manual. FAO Fisheries technical paper 306-1, rev. 2.

G.J. de Graaf is from the Nefisco Foundation, Lijnbaansgracht 14 C, 1015 GN Amsterdam, the Netherlands.
Email: degraaf@nefico.org
P. Dekker is from Xi Consultancy, P.O. Box 1000, 2600 BA Delft, the Netherlands.
Email: Dekker@Xi-advies.nl